Maple-assisted verification of empirical formulas for rows and columns of A245869

Robert Israel

06 Aug 2024

Consider the last two entries of an array as its "state". These fall into the following classes:

(a) $(x, k-x)$ where $x \neq k-x$, i.e. $x \neq \frac{k}{2}$.

(b) (x, y) where $x+y \neq k$, $x \neq y$, and neither x nor y is $\frac{k.}{2}$.

(c) (x, x) where $x \neq \frac{k}{2}$.

(d) $\left(x, \frac{k}{2}\right)$ where $x \neq \frac{k}{2}$.

(e) $\left(\frac{k}{2}, x\right)$ where $x \neq \frac{k}{2}$.

(f) $\left(\frac{k}{2}, \frac{k}{2}\right)$.

The last three states are only present when k is even. We'll first consider the case where k is odd.

If we have an allowed array of length n+2 with state (a), then we can extend it to an array of length $n+3$ of state (a) in one way, by appending x, or an array of state (c) by appending $k-x$, or an array of state (b) by appending any of the other $k-1$ possible elements.

An allowed array of state (b) can go to state (a) by appending $k-y$, or (b) by appending $k-x$.

An array of state (c) can go to (a) by appending $k-x$. No other elements can be appended without violating the constraint that some pair adds to k.

Thus if $v(n)$ is the column vector whose entries are the numbers of allowed arrays of length n in states 1 to 3, we have $v(n+1) = M\, v(n)$ where M is the transition matrix

$$M = \begin{bmatrix} 1 & 1 & 1 \\ k-1 & 1 & 0 \\ 1 & 0 & 0 \end{bmatrix}$$

Taking $v(0) = \begin{bmatrix} k+1 \\ k^2-1 \\ k+1 \end{bmatrix}$ and $e = \begin{bmatrix} 1 \\ 1 \\ 1 \end{bmatrix}$, we have $T(n,k) = e^T v(n) = e^T M^n v(0)$. To check, here are $T(n,5)$, the fifth column of the table, for n from 1 to 10.

```
> e:= Vector[column](3,1): v0:= <k+1,k^2-1,k+1>: M:= <<1,k-1,1>|<1,
  1,0>|<1,0,0>>:
  M5:= eval(M,k=5): v5:= eval(v0,k=5):
  seq(e^%T . M5^n . v5, n=1..10);
```

$$90, 318, 960, 3102, 9726, 30900, 97602, 309078, 977664, 3094038 \quad (1)$$

Now it turns out that the vectors $v(0)$, $v(1)$, $v(2)$ and $v(3)$ are linearly dependent. To see this, we form a matrix with these vectors as columns and find its rank and null space.

```
> Q:= map(normal,<M^3 . v0| M^2 . v0| M . v0| v0>)
```

$$Q := \begin{bmatrix} k^3+7k^2+4k-2 & 3k^2+3k & k^2+2k+1 & k+1 \\ 4k^3+3k^2-4k-3 & k^3+3k^2-k-3 & 2k^2-2 & k^2-1 \\ 3k^2+3k & k^2+2k+1 & k+1 & k+1 \end{bmatrix} \quad (2)$$

```
> with(LinearAlgebra): Rank(Q);
```

$$3 \quad (3)$$

```
> NullSpace(Q);
```

$$\left\{ \begin{bmatrix} 1 \\ -2 \\ -k+1 \\ 1 \end{bmatrix} \right\} \quad (4)$$

This says that $v(3) - 2\,v(2) - (k-1)\,v(1) + v(0) = 0$, which implies the recurrence for odd k: $T(n+3,k) - 2\,T(n+2,k) - (k-1)\,T(n+1,k) + T(n,k) = 0$.

Now consider the case where k is even, and we have 6 states to consider.
From state (a), you could get to (a) by appending x, or (b) by appending anything other than x, $k-x$ or $\frac{k}{2}$, or (c) by appending $k-x$. or (d) by appending $\frac{k}{2}$.
From state (b), you could get to (a) by appending $k-y$, or (b) by appending $k-x$.
From state (c), you could get to (a) by appending $k-x$.
From state (d), you could get to (e) by appending $k-x$, or (f) by appending $\frac{k}{2}$.
From state (e), you could get to (a) by appending $k-x$, or (d) by appending $\frac{k}{2}$.
From state (f), you could get to (e) by appending anything other than $\frac{k}{2}$, or (f) by appending $\frac{k}{2}$.
Thus in this case the matrix is

```
> M:= <<1,k-2,1,1,0,0>|<1,1,0,0,0,0>|<1,0,0,0,0,0>|<0,0,0,0,1,
  1>|<1,0,0,1,0,0>|<0,0,0,0,k,1>>;
```

$$M := \begin{bmatrix} 1 & 1 & 1 & 0 & 1 & 0 \\ k-2 & 1 & 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 & 0 & 0 \\ 1 & 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 1 & 0 & k \\ 0 & 0 & 0 & 1 & 0 & 1 \end{bmatrix} \quad (5)$$

and the initial vector is

```
> v0:= <k,k*(k-2),k,k,k,1>;
```

$$v0 := \begin{bmatrix} k \\ k(k-2) \\ k \\ k \\ k \\ 1 \end{bmatrix} \quad (6)$$

To check, we'll compute $T(n, 6)$ for n from 1 to 10.

```
> M6:= eval(M,k=6): v6:= eval(v0,k=6): e:= Vector(6,1):
  seq(e^%T . M6^n . v6, n=1..10);
```

$$127, 493, 1579, 5515, 18505, 63241, 214315, 729097, 2475985, 8415217 \quad (7)$$

```
> Q:= <M^6 . v0 | M^5 . v0 | M^4 . v0 | M^3 . v0 | M^2 . v0 | M .
  v0 | v0>:
  Rank(Q);
```

$$6 \quad (8)$$

```
> NullSpace(Q);
```

(9)

$$\left\{\left[\begin{array}{c} -\frac{1}{k-1} \\ \frac{3}{k-1} \\ \frac{k-3}{k-1} \\ -\frac{1}{k-1} \\ -\frac{2k-3}{k-1} \\ -\frac{k(k-2)}{k-1} \\ 1 \end{array}\right]\right\} \qquad \textbf{(9)}$$

Thus we find that
$-v(6) + 3\, v(5) + (k-3)\, v(4) - v(3) + (2\,k-3)\, v(2) - k(k-2)\, v(1) + (k-1)\, v(0) = 0,$
which implies the recurrence for even k:

$$-T(n+6,k) + 3\, T(n+5,k) + (k-3)\, T(n+4,k) - T(n+3,k) + (2\,k-3)\, T(n+2,k) - k(k-2)\, T(n+1,k) + (k-1)\, T(n,k) = 0$$

.